/*
Laura Stoker	
Proposal title: "Equivalency Framing of Societal Problems and Policy Solutions"


HYPOTHESES

Stated-Hyp1: "We hypothesize that the public will express more concern about societal problems when informed about the incidence of bad outcomes (vs. good)." (p. 1)
	
	* See note below about DVs. I'm assuming concern means anger and worry

	Test-Hyp1: The level of anger about the problem will be higher among those exposed to bad vs. good outcomes.

	Test-Hyp2: The level of worry about the problem will be higher among those exposed to bad vs. good outcomes.	
	
Stated-Hyp2: "We hypothesize that the public... will give more support to ameliorative policies described as reducing the incidence of negative outcomes (vs. increasing the incidence of good ones)." (p. 1)

	Test-Hyp3: Willingness to sign a petition will be higher among those exposed to ameliorative vs. enhancing policies.

	Test-Hyp4: Policy approval will be higher among those exposed to ameliorative vs. enhancing policies.
	
	Test-Hyp5: Willingness to volunteer will be higher among those exposed to ameliorative vs. enhancing policies.
	
********************************************************************************
NOTES:

- Policy statements vary by amount of loss, cost, etc (page 4). However, there 
is no hypothesis about these variations.

- dependent variable measures are described on page 4:

"Our dependent variable questions, which follow the problem statement, ask respondents to 
(1) report their level of anger and worry about the problem; 
(2) judge the seriousness of the problem; and 
(3) determine the priority the government should give to the problem."

On page 5 there is a slightly different list: 
"Our dependent variable measures... ask for:
(1) an overall opinion of the policy initiative, 
(2) willingness to support it by signing a petition,
(3) willingness to join a volunteer effort that is called for by the policy initiative."

- We are using the DVs that pertain to the stated hypotheses.
*/

clear all
use "Stoker1063.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

	* bad vs. good scenario (see quex)
	tab DOV_COND
	recode DOV_COND (1=1) (2=0) (3=1) (4=0) (5=1) (6=0) (7=1) (8=0) (9=1) (10=0) (11=1) (12=0) (13=1) (14=0) (15=1) (16=0),  gen(scenario_bad)
	tab scenario_bad
	
	* flag for whether policy solution showed 
	gen policysolution=1 if DOV_COND>8
	
	* policy solution is ameliorative vs. enhancing
		// ameliorative policies are matched with negative scenarios and enhancing policies are matched with good scenarios in all cases.
	gen ameliorative=scenario_bad if policysolution==1

	
* OUTCOME MEASURES

	* worry
		tab WORRY 
		replace WORRY =. if WORRY>7

		// note: worry sums up worry 1 and worry 0 (treatment and control)
	 
	* anger
		tab ANGER
		replace ANGER =. if ANGER>7
		
	* willingness to sign a petition
		tab PETITION
		replace PETITION=. if PETITION>7
		
	* approval of policy solution
		tab OPINION
		replace OPINION=. if OPINION>7
		
	* willingness to volunteer to support policy effort
		tab VOLUNTEER
		replace VOLUNTEER=. if VOLUNTEER>7

	
********************************************************************************

* ANALYSIS
	

*Test-Hyp1: The level of anger about the problem will be higher among those exposed to bad vs. good outcomes.

	reg ANGER i.scenario_bad
	// do not reject. 0.000
	tess 1.scenario_bad +, init(Stoker1063) bonf(2)

*Test-Hyp2: The level of worry about the problem will be higher among those exposed to bad vs. good outcomes.	
	
	reg WORRY i.scenario_bad
	// do not reject. 0.000
	tess 1.scenario_bad +, bonf(2)
	
*Test-Hyp3: Willingness to sign a petition will be higher among those exposed to ameliorative vs. enhancing policies.
	
	reg PETITION i.ameliorative
	// reject. 0.873
	tess 1.ameliorative +, bonf(3)
	
*Test-Hyp4: Policy approval will be higher among those exposed to ameliorative vs. enhancing policies.

	reg OPINION i.ameliorative
	//reject. 0.776
	tess 1.ameliorative +, bonf(3)
	
*Test-Hyp5: Willingness to volunteer will be higher among those exposed to ameliorative vs. enhancing policies.

	reg VOLUNTEER i.ameliorative
	// reject. 0.352
	tess 1.ameliorative +, bonf(3)
